Behavior Research Methods & Instrumentation
1983, 15 (4), 465-466

PROGRAM ABSTRACTS/ALGORITHMS

A versatile sorting and ranking program

ALFRED L. BROPHY
Behavioral Science Associates, West Chester, Pennsylvania

McNally and Wood (1982) recently provided a BASIC program that assigns ranks to a set of scores. This short program maintains the original order of the scores and correctly ranks tied scores. It can be a useful routine in many programs that calculate statistics based on ranks.

Table 1 presents SORTRANK, another BASIC program to sort and rank scores, one that has proved to be better for some applications. This program is versatile in that it produces two types of output: (1) scores in sorted order, together with ranks and associated subject identification numbers, and (2) ranks of the scores in their original order. Although the second type of output is often desired, the first type is sometimes preferable or necessary, for example, in calculating Kendall's tau.

SORTRANK uses an insertion sort (Lines 50-70) to sort the scores (Array X), along with their subject numbers (Array Y). Lines 80-100 then check whether consecutive sorted scores are tied. If there is a tie, the program branches to Lines 110-140, which check for further ties and calculate the ranks of the tied scores. Array R contains the ranks. Next, Line 150 determines the rank corresponding to each subject number from 1 to N, thus creating an array of ranks (Array Z) corresponding to the scores in their original order. Finally, Lines 160-200 print the results available from the program.

Execution Time. At the cost of somewhat greater length, SORTRANK executes more quickly than McNally and Wood's (1982) program. The sorting method compares favorably in speed with other sorting algorithms for lists of up to about 50 items. In the worst case, in which the scores are input in reverse order, the main part of the program (prior to the resolution of ties) requires $N(N - 1)/2$ comparisons. In checking for ties, the program makes $(N - 1)$ comparisons. In contrast, each analogous part of McNally and Wood's program always requries N^2 comparisons.

Time trials were run on a TRS-80 Model I, and the results reported are the means of five trials for each set of data. When 20 untied scores were input in reverse order, SORTRANK executed in 7 sec, compared with 15 sec for McNally and Wood's (1982) program. For 20 scores consisting of the same value (an N-way tie), the times were 2 and 17 sec, respectively. (Before McNally and Wood's program was used, two misprints were corrected: the first "+" in Line 50 was changed to "=," and the last comma in Line 70 was deleted. To increase speed, the variables I, J, and N were defined as integers.)

The author's mailing address is: Behavioral Science Associates, P.O Box 748, West Chester, Pennsylvania 19380.

Table 1
SORTRANK Program to Sort and Rank an Array of Numerical Data

```
SORTRANK Program to Sort and Rank an Array of Numerical Data

10  DEFINT I-N, S, Y
20 INPUT"Enter N "; N: N1=N-1
30 DIM X(N), Y(N), R(N), Z(N)
40 FOR I=1 TO N: PRINT"Score # " I;: INPUT X(I): Y(I)=I: R(I)=I:
   NEXT I
50 FOR I=1 TO N1: T=X(I+1): TT=Y(I+1): FOR J=I TO 1 STEP -1
60 IF T<X(J) THEN X(J+1)=X(J): Y(J+1)=Y(J): NEXT J: J=0
70 X(J+1)=T: Y(J+1)=TT: NEXT I
80 K=1
90 FOR I=K TO N1: IF X(I)=X(I+1) THEN 110
100 NEXT I: GOTO 150
110 M=2: K=I+1: S=I+K: IF K>N1 THEN 130
120 FOR J=K TO N1: IF X(J)=X(J+1) THEN M=M+1: S=S+J+1: NEXT J
130 R=S/M: FOR J=I TO I+M-1: R(J)=R: NEXT J
140 K=J: IF K<N THEN 90
150 FOR I=1 TO N: Z(Y(I))=R(I): NEXT I
160 PRINT"Data Sorted by Score:": PRINT"Score","Subject #","Rank"
170 FOR I=1 TO N: PRINT X(I), Y(I), R(I): NEXT I
180 PRINTTAB(35)"Press <ENTER> to continue";: INPUT E$
190 PRINT"Rank Corresponding to Subject:": PRINT"Subject #","Rank"
200 FOR I=1 TO N: PRINT I, Z(I): NEXT I
210 END
```

When a large number of scores is to be sorted, Singleton's (1969) procedure is recommended. This highly efficient algorithm has recently been incorporated in the SPSS nonparametric correlation program and has substantially improved the speed of that program (Hull & Nie, 1981). Table 2 shows a BASIC version of Singleton's procedure, which was originally published in FORTRAN and ALGOL versions, with additions necessary to handle subject numbers. The listing in Table 2 can be used in place of Lines 50-70 in SORTRANK. With this substitution, the program sorted and ranked 100 randomly arranged scores in 39 sec, compared with 78 sec for SORTRANK and 355 sec for McNally and

Table 2
BASIC Coding of Singleton's (1969) Sorting Algorithm

```
BASIC Coding of Singleton's (1969) Sorting Algorithm

50 DIM IU(11), IL(11): M=1: I=1: J=N
51 IF I>=J THEN 63
52 K=I: IJ=(I+J)/2: T=X(IJ): T1=Y(IJ)
53 IF X(I)>T THEN X(IJ)=X(I): X(I)=T: T=X(IJ): Y(IJ)=Y(I):
   Y(I)=T1: T1=Y(IJ)
54 L=J: IF X(J)>=T THEN 57
55 X(IJ)=X(J): X(J)=T: T=X(IJ): Y(IJ)=Y(J): Y(J)=T1: T1=Y(IJ)
56 IF X(I)>T THEN X(IJ)=X(I): X(I)=T: T=X(IJ): Y(IJ)=Y(I):
   Y(I)=T1: T1=Y(IJ)
57 L=L-1: IF X(L)>T THEN 57
58 TT=X(L): T2=Y(L)
59 K=K+1: IF X(K)<T THEN 59
60 IF K<=L THEN X(L)=X(K): X(K)=TT: Y(L)=Y(K): Y(K)=T2: GOTO 57
61 IF L-I<=J-K THEN IL(M)=K: IU(M)=J: J=L: M=M+1: GOTO 65
62 IL(M)=I: IU(M)=L: I=K: M=M+1: GOTO 65
63 M=M-1: IF M=0 THEN 80 ELSE I=IL(M): J=IU(M)
65 IFJ>I THEN 52
66 IF I=1 THEN 51 ELSE I=I-1
67 I=I+1: IF I=J THEN 63
68 T=X(I+1): T1=Y(I+1): IF X(I)<=T THEN 67 ELSE K=I
69 X(K+1)=X(K): Y(K+1)=Y(K): K=K-1: IF T<X(K) THEN 69
70 X(K+1)=T: Y(K+1)=T1: GOTO 67
```

Wood's (1982) program. For 250 randomly arranged scores, the times were 108, 453, and 2,213 sec, respectively.

Language. SORTRANK is written in TRS-80 Level II BASIC, and it will run on TRS-80 Model I, Model II, and Model III computers. Changes necessary for other BASIC dialects may include the deletion of the DEFINT statement, the insertion of LET in assignment statements, and recoding to avoid multiple-statement lines and ELSE statements. SORTRANK is 634 bytes long without spaces; with Singleton's (1969) algorithm, the program is 1,198 bytes long. The memory required to run the program is a function of the number of scores. With 100 floating-point scores, including ties, the program with Singleton's algorithm runs in approximately 2.7 KB on the Model I.

Availability. A listing of the program can be obtained at no cost from Alfred L. Brophy, Behavioral Science Associates, P.O. Box 748, West Chester, Pennsylvania 19380. For a TRS-80 Model I disk copy of the program, enclose a system diskette and envelope with return postage.

REFERENCES

HULL, C. H., & NIE, N. H. (EDS.). *SPSS update 7-9: New procedures and facilities for releases 7-9.* New York: McGraw-Hill, 1981.

McNALLY, K. A., & WOOD, D. L. BASIC rank-order program. *Behavior Research Methods & Instrumentation*, 1982, **14**, 545.

SINGLETON, R. C. Algorithm 347: An efficient algorithm for sorting with minimal storage. *Communications of the ACM*, 1969, **12**, 185-187.

(Manuscript accepted for publication May 13, 1983.)